407 CS 538 Spring 2002^©

Backtracking

Ifwereacha pointwherea goalcan’t be matched, or the body of a rule can’tbematched,webacktrackto the last(mostrecent)spotwhereachoice of matching a particular fact or rule was made. We then try to match a different fact or rule. If this can’t be done,wegobackto thenextprevious place where a choice was made and try a different match there. We try alternativesuntilweareableto solve all the goals in our query or until all possible choices have been tried and found to fail. If this happens, we answer “no” the query can’t be solved. Aswetryto matchfactsandruleswe try them in their order of definition. 408 CS 538 Spring 2002^©

Example

Let’s trace how

| ?- grandMotherOf(tom,GM).

is solved. Recall that

grandMotherOf(X,GM) :-
  motherOf(X,M),
     motherOf(M,GM).

grandMotherOf(X,GM) :-
  fatherOf(X,F),
    motherOf(F,GM).
fatherOf(tom,dick).
fatherOf(dick,harry).
fatherOf(jane,harry).
motherOf(tom,judy).
motherOf(dick,mary).
motherOf(jane,mary).

409 CS 538 Spring 2002^©

We try the firstgrandMotherOf rule first. This forcesX = tom. We have to solve

 motherOf(tom,M),
     motherOf(M,GM).

We now try to solve

motherOf(tom,M)

This forcesM = judy. We then try to solve

motherOf(judy,GM)

None of themotherOf rules match this goal, so we backtrack. No other

motherOf rule can solve
motherOf(tom,M)

so we backtrack again and try the secondgrandMotherOf rule:

410 CS 538 Spring 2002^©

grandMotherOf(X,GM) :-
  fatherOf(X,F),
    motherOf(F,GM).

This matches, forcingX = tom. We have to solve

fatherOf(tom,F), motherOf(F,GM).

We can match the first goal with

fatherOf(tom,dick).

This forcesF = dick. We then must solve

motherOf(dick,GM)

which can be matched by

motherOf(dick,mary).

We havematchedallourgoals, sowe know the query is true, with

GM = mary.

411 CS 538 Spring 2002^©

List Processing in Prolog

Prologhasa notationsimilarto “cons cells” of Lisp and Scheme. The “.” functor (predicate name) acts like cons. Hence.(a,b) inPrologisessentially the same as(a . b) in Scheme. Lists in Prolog are formed much the same way as in Scheme and ML:

[] is the empty list

[1,2,3] is an abbreviation for
 .(1, .(2, .(3,[])))

just as

(1,2,3) inSchemeisanabbreviation for

(cons 1 (cons 2 (cons 3 () )))

412 CS 538 Spring 2002^©

The notation[H|T] represents a list withH matching the head of the list andT matching the rest of the list. Thus[1,2,3] ≡ [1| [2,3]] ≡

[1,2| [3]] ≡ [1,2,3| [] ]

As in ML, “_” (underscore) can be used as a wildcard or “don’t care” symbol in matches. Given the fact

 p([1,2,3,4]).

The query

 | ?- p([X|Y]).

answers

X = 1,
Y = [2,3,4]

413 CS 538 Spring 2002^©

The query

p([_,_,X|Y]).

answers

X = 3,
Y = [4]

414 CS 538 Spring 2002^©

List Operations in Prolog

Listoperationsaredefinedusingrules and facts. The definitions are similar to those used in Scheme or ML, but they arenon-procedural. That is you don’t given an execution order.Instead,yougiverecursiverules and non-recursive “base cases” that characterize the operation you are defining. Considerappend:

 append([],L,L).
 append([H|T1],L2,[H,T3]) :-
   append(T1,L2,T3).

The first fact says that an empty list (argument 1) appended to any listL (argument2) givesL (argument3) as its answer.

415 CS 538 Spring 2002^©

Theruleinline2 saysthat ifyoutake a listthat beginswithH andhasT1 as the restof the listandappendit to a listL thenthe resultingappendedlist will begin withH. Moreover, the rest of the resulting list,T3, is the result of appendingT1 (therestofthe first list)withL2 (the second input list). The query

 | ?- append([1],[2,3],[1,2,3]).

answers

Yes

becausewithH=1, T1=[], L2 =[2,3] andT3=[2,3] it must be the case that append([],[2,3],[2,3]) is true and fact (1) says that this is so.

416 CS 538 Spring 2002^©

Inverting Inputs and Outputs

In Prolog the division between “inputs” and “outputs” is intentionally vague. We can exploit this. It is often possible to “invert” a query and ask whatinputs would compute a given output. Few other languages allow this level of flexibility. Consider the query

append([1],X,[1,2,3]).

This asks Prolog to find a listX such that if we append[1] to X we will get [1,2,3]. Prolog answers

X = [2,3]

417 CS 538 Spring 2002^©

How does it choose this answer? First Prolog tries to match the query against fact (1) or rule (2). Fact (1) doesn’t match (the first arguments differ) so we match rule (2). This gives usH=1, T1=[], L2=X and

T3 = [2,3].

We next have to solve the body of rule (2) which is

append([],L2,[2,3]).

Fact (1) matches this, and tells us that L2=[2,3]=X, and that’s our answer!

418 CS 538 Spring 2002^©

The Member Relation

A common predicate when manipulating lists is a membership test—is a given value a member of a list? An “obvious” definition is a recursive one similar to what we might program in Scheme or ML:

member(X,[X|_]).
member(X,[_|Y]):- member(X,Y).

This definition states that the first argument,X, is a member of the secondargument(a list)if X matches the head of the listor if X is (recursively) a member of the rest of the list.

419 CS 538 Spring 2002^©

Note that we don’t have to “tell” Prologthat X can’tbea memberofan empty list—if we don’t tell Prolog that something is true, it automaticallyassumesthat it mustbe false. Thus saying nothing about membership in an empty list is the sameassayingthat membershipinan empty list is impossible. Sinceinputsandoutputsina relation are blurred, we can usemember in an unexpectedway—toiteratethrougha list of values.

420 CS 538 Spring 2002^©

Ifwewantto knowif anymemberof a listL satisfies a predicatep, we can simply write:

member(X,L),p(X).

There is no explicit iteration or searching. We simply ask Prolog to find anX such thatmember(X,L) is true (X is inL) andp(X) is true. Backtracking will find the “right” value forX (if any suchX exists). This is sometimes called the “guess and verify” technique. Thus we can query

member(X,[3,-3,0,10,-10]),
  (X > 0).

This asks for anX in the list

[3,-3,0,10,-10] which is greater

than 0. Prolog answers

421 CS 538 Spring 2002^©

X = 3 ;

X = 10 ;

Note too that our “obvious” definition of member is not the only one possible. Analternativedefinition(whichisfar less obvious) is

member(X,L) :-
  append(_,[X|_],L).

This definition saysX is a member of L if Icantakesomelist(whosevalue Idon’tcareabout)andappendit to a list that begins withX (and which ends with values I don’t care about) and get a list equal toL. Said more clearly, X is a member ofL if X is anywhere in the “middle” ofL.

422 CS 538 Spring 2002^©

Prolog solves a query involving

member by partitioning the listL in

all possible ways, and checking to see if X ever is the head of the second list.Thusformember(X,[1,2,3]), it tries the partition[] and [1,2,3] (exposing1 asa possibleX),then[1] and[2,3] (exposing2) and finally

[1,2] and [3] (exposing3).

423 CS 538 Spring 2002^©

Sorting Algorithms

Sortingalgorithmsaregoodexamples ofProlog’sdefinitionalcapabilities.In a Prolog definition the “logic” of a sorting algorithm is apparent, strippedofthecumbersomedetailsof datastructuresandcontrolstructures that dominate algorithms in other programming languages. Consider the simplest possible sort imaginable, which we’ll call the “naive sort.” Atthesimplestlevela sortingofa list

L requires just two things:
• The sorting is a permutation (a

reordering) of the values inL.

• The values are “in order” (ascending or descending). 424 CS 538 Spring 2002^©

We can implement this concept of a sort directly in Prolog. We (a) permute an input list (b) check if it is in sorted order (c) repeat (a) & (b) until a sorting is found.

425 CS 538 Spring 2002^©

Permutations

Let’s first look at how permutations are defined in Prolog. In most languagesgeneratingpermutationsis non-trivial—youneeddata structures to store the permutations you are generating and control structures to visit all permutations in some order. In Prolog, permutations are defined quite concisely, though with a bit of subtlety:

perm(X,Y) will be true if listY is a

permutation of listX. Only two definitions are needed:

perm([],[]).
perm(L,[H|T]) :-
 append(V,[H|U],L),
 append(V,U,W), perm(W,T).

426 CS 538 Spring 2002^©

The first definition,

perm([],[]).

is trivial. An empty list may only be permuted into another empty list. The second definition is rather more complex:

perm(L,[H|T]) :-
append(V,[H|U],L),
 append(V,U,W), perm(W,T).

This rule says a listL may be permutedinto a listthat beginswith H and ends with listT if:

(1) L may be partitioned into two lists, V and [H|U]. (That is, H is somewhere in the “middle” ofL). (2) ListsV and U (all ofL exceptH) may be appended into listW. (3) ListW may be permuted intoT.

427 CS 538 Spring 2002^©

Let’s seeperm in action:

| ?- perm([1,2,3],X).
X = [1,2,3] ;

X = [1,3,2] ;

X = [2,1,3] ;

X = [2,3,1] ;

X = [3,1,2] ;

X = [3,2,1] ;
no

We’lltracehowthe first fewanswers are computed. Note though thatall permutationsaregenerated,andwith no apparent data structures or control structures. We start withL=[1,2,3] and

X=[H|T].

We first solveappend(V,[H|U],L), which simplifies to

append(V,[H|U],[1,2,3]).

428 CS 538 Spring 2002^©

One solution to this goal is

V = [], H = 1, U = [2,3]

Wenextsolveappend(V,U,W) which simplifies toappend([],[2,3],W). The only solution for this isW=[2,3]. Finally, we solveperm(W,T), which simplifies toperm([2,3],T). One solution to this isT=[2,3]. This gives us our first solution:

[H|T]=[1,2,3].

Togetournextsolutionwebacktrack. Where is the most recent place we madea choiceofhowto solvea goal? It was atperm([2,3],T). We chose

T=[2,3], but T=[3,2] is another

solution. Using this solution, we get out next answer[H|T]=[1,3,2].

429 CS 538 Spring 2002^©

Let’s try one more. We backtrack again.Nomoresolutionsarepossible for perm([2,3],T), sowebacktrack to an earlier choice point. Atappend(V,[H|U],[1,2,3]) another solution is

V=[1], H = 2, U = [3]

Using this binding, we solve

append(V,U,W) which simplifies to
append([1],[3],W).Thesolutionto

this must beW=[1,3]. We then solveperm(W,T) which simplifies toperm([1,3],T). One solution to this isT=[1,3]. This makesourthirdsolutionfor[H|T] =

[2,1,3].

Youcancheckout the otherbindings that lead to the last three solutions.

430 CS 538 Spring 2002^©

A Permutation Sort

Now that we know how to generate permutations, the definition of a permutation sort is almost trivial. We define aninOrder relation that characterizes our notion of when a list is properly sorted:

inOrder([]).
inOrder([_]).
inOrder([A,B|T]) :-
   A =< B, inOrder([B|T]).

Thesedefinitionsstatethat a nulllist, and a list with only one element are always in sorted order. Longer lists are in order if the first two elements are in proper order. (A=<B) checks this and then the rest of the list, excluding the first element, is checked.

431 CS 538 Spring 2002^©

Now our naive permutation sort is only one line long:

naiveSort(L1,L2) :-
  perm(L1,L2), inOrder(L2).

And the definition works too!

| ?-
naiveSort([1,2,3],[3,2,1]).
no
?- naiveSort([3,2,1],L).
L = [1,2,3] ;
no
| ?-
naiveSort([7,3,88,2,1,6,77,
 -23,5],L).
L = [-23,1,2,3,5,6,7,77,88]

432 CS 538 Spring 2002^©

Though this sort works, it is hopelessly inefficient—it repeatedly “shuffles” the input until it happens to findanorderingthat issorted.The process is largely undirected. We don’t“aim”towardacorrectordering, but just search until we get lucky.